

# Modeling animal populations

## The basic exponential model of population growth

The simplest model of population growth is the exponential model. The assumption is that the population is always growing at a rate equal to fixed percentage of the population (so the “growth rate” is a constant). This is represented as

$$\frac{dP}{dt} = r P$$

where  $r$  is the growth rate. This is a very easy model to solve assuming you know the starting population  $p_0$  -


```
DSolveValue[ {p[0] == p0, p'[t] == r p[t]}, p[t], t]
```

$e^{r t} p_0$

So if your population started at 300 (million) with a growth rate of .02 then the formula for the population would be  $300 e^{.02 t}$  and the graph would look like this:

```
Plot[ 300 Exp[.02 t], {t, 0, 100}, AxesLabel -> {"time", "population in millions"}]
```

population in millions



This is not very realistic as it there are no limits on growth. Suppose you wanted to find when there would be  $10^{25}$  people - you would simply set  $300 e^{.02 t} = 10^{25} / 10^6$  (dividing by  $10^6$  as the units are millions of people):

```
Solve[ 300 Exp[.02 t] == 10^25 / 10^6, t] // Quiet
```

```
{{t -> 1902.27}}
```

So after a bit less than 1902 years and 3 months there would be  $10^{25}$  people. I didn't pick the number  $10^{25}$  at random - the mass of the earth in kilograms is about  $5.94 \times 10^{24}$ , so there would be more people than kilograms! This isn't very realistic - in practice as the number of people increases limits on available food, water, and energy would cut into the birth rate, undercutting the assumption that the rate was always 2%.

## The logistic model with a threshold

One common population model (the logistic model with a threshold) incorporates tries to remedy the defects in the previous model by adding two additional features:

- 1) There is a viability threshold  $V$ ; below this value the population should die off.
- 2) There is an environmental “carrying capacity”  $C$  - a limit to how big a population can get - over this value you

should get some die off.


The equation that models the population  $P$  in this case is

$$\frac{dP}{dt} = rP\left(\frac{P}{V} - 1\right)\left(1 - \frac{P}{C}\right)$$

This equation is not easily solvable for general values of  $r$ ,  $V$ , and  $C$ , but we can still investigate what the solution looks like using `Manipulate` to put in specific values of  $r$ ,  $V$ , and  $C$ :

## Logistic with threshold behavior

```
Manipulate[Module[{equation1},
  equation1 = NDSolveValue[{y'[t] == -r y[t] (1 - y[t] / c) (1 - y[t] / v),
    y[0] == y0, WhenEvent[y[t] < 0, y[t] == 0]}, y, {t, 0, 100}];
  Plot[{Evaluate[equation1[t]], c, v}, {t, 0, 100},
    PlotRange -> {0, c + 5}, PlotLabel -> "Population over time",
    PlotLegends -> {"Population", "Carrying Capacity", "Viability Threshold"}],
  {{v, 3, "Viability Threshold: "}}, {{c, 10, "Carrying Capacity: "}},
  {{r, .05, "Growth Rate: "}}, {{y0, c / 2, "Initial population"}, 0, c + 5, .01},
  LocalizeVariables -> True, SaveDefinitions -> True]
```



By changing the value for the initial population you can see this model behaves very differently than the exponential model:

1) If the initial population is below the viability threshold  $V$  then the population dies down to 0 (which agrees with the original design of the model).

- 2) If the initial population is between the viability threshold  $V$  and the carrying capacity  $C$  then the population will increase to  $C$ , leveling off to  $C$  as time progresses.
- 3) If the initial population is above the carrying capacity  $C$  then population will drop to  $C$  in the long run.

## Modifying the model to account for hunting or fishing

The logistic with threshold model assumes that there's no outside influence on the population. But what if you allow people to hunt or fish? If you license the hunters then you can assume that they will take out animals at a constant rate  $k$  this will change the equation from

$$\frac{dP}{dt} = r P \left( \frac{P}{V} - 1 \right) \left( 1 - \frac{P}{C} \right)$$

to

$$\frac{dP}{dt} = r P \left( \frac{P}{V} - 1 \right) \left( 1 - \frac{P}{C} \right) - k$$


Like the previous model this is too difficult to solve for arbitrary values of the parameters, but we can use Manipulate to investigate the behavior. This manipulation will allow you to vary all the parameters including the new hunting/fishing parameter  $k$ .

## The modified behavior

```

Manipulate[ Module[{equation1, equation2},
  equation1 = NDSolveValue[{y'[t] == -r y[t] (1 - y[t] / c) (1 - y[t] / v),
    y[0] == y0, WhenEvent[y[t] < 0, y[t] == 0]} , y, {t, 0, tmax}];
  equation2 = NDSolveValue[{y'[t] == -r y[t] (1 - y[t] / c) (1 - y[t] / v) - k,
    y[0] == y0, WhenEvent[y[t] < 0, y[t] == 0]} , y, {t, 0, tmax}];
  Plot[ {Evaluate[equation1[t]], equation2[t], c, v},
    {t, 0, tmax}, PlotRange -> {0, c + 5},
    PlotLegends -> {"Logistic", "Logistic with fishing", "Carrying Capacity",
      "Viability Threshold"}, PlotLabel -> "Population over time"],
  {{v, 3, "Viability Threshold: "}}, {{c, 10, "Carrying Capacity: "}},
  {{r, .05, "Growth Rate: "}},
  {{k, 0.02, "Fishing rate (per year)"}, 0, .2, .001},
  {{y0, c / 2, "Initial population"}, 0, c + 5, .01},
  {{tmax, 100, "Maximum time shown"}},
  LocalizeVariables -> True, SaveDefinitions -> True]

```



Like the previous model if your initial population is below the viability threshold then your population will automatically die off. For values above the threshold things change, depending on the value of the hunting/fishing parameter  $k$ ; for low values of  $k$  the population will stabilize (although at lower level than the original “logistic” model) but for higher values of  $k$  the population will drop below the viability threshold and die off

entirely. In fact sometimes a very tiny change in the value of  $k$  will switch from a “stable” mode to a “die-off” mode, which shows the importance of setting catch levels appropriately.

Author: Dr. Christopher Moretti

Revision Date: August 3rd, 2015.